


Programming Languages and Compilers (CS 421)



Dennis Griffith
0207 SC, UIUC

<http://www.cs.uiuc.edu/class/cs421/>

Based in part on slides by Mattox Beckman, as updated by Vikram Adve, Gul Agha, and Elsa Gunter



Folding

```
# let rec fold_left f a list = match list
  with [] -> a | (x :: xs) -> fold_left f (f a x) xs;;
val fold_left : ('a -> 'b -> 'a) -> 'a -> 'b list -> 'a =
  <fun>
```

```
fold_left f a [x1; x2;...;xn] = f(...(f (f a x1) x2)...)xn
```

```
# let rec fold_right f list b = match list
  with [ ] -> b | (x :: xs) -> f x (fold_right f xs b);;
val fold_right : ('a -> 'b -> 'b) -> 'a list -> 'b -> 'b =
  <fun>
```

```
fold_right f [x1; x2;...;xn] b = f x1(f x2 (...(f xn b)...))
```



Folding - Tail Recursion

```
- # let rev list =  
-   fold_left  
-   (fun r -> fun x -> x :: r) //comb op  
-   [] //accumulator cell  
-   list
```



Folding

- Can replace recursion by `fold_right` in any forward primitive recursive definition
 - Primitive recursive means it only recurses on immediate subcomponents of recursive data structure
- Can replace recursion by `fold_left` in any tail primitive recursive definition



Map from Fold

```
# let map f list =  
  fold_right (fun x y -> f x :: y) list [ ];;  
val map : ('a -> 'b) -> 'a list -> 'b list =  
  <fun>  
# map ((+)1) [1;2;3];;  
- : int list = [2; 3; 4]
```

- Can you write `fold_right` (or `fold_left`) with just `map`? How, or why not?



Map from Fold

```
# let map f list =  
  fold_right (fun x y -> f x :: y) list [ ];;  
val map : ('a -> 'b) -> 'a list -> 'b list =  
  <fun>  
# map ((+)1) [1;2;3];;  
- : int list = [2; 3; 4]
```

■ Can you write `fold_right` (or `fold_left`) with just `map`? How, or why not?

■ `fold_right (fun x a -> (f x) :: a) list []`



Higher Order Functions

- A function is *higher-order* if it takes a function as an argument or returns one as a result
- Example:

```
# let compose f g = fun x -> f (g x);;
```

```
val compose : ('a -> 'b) -> ('c -> 'a) -> 'c -> 'b = <fun>
```

- The type `('a -> 'b) -> ('c -> 'a) -> 'c -> 'b` is a higher order type because of `('a -> 'b)` and `('c -> 'a)` and `-> 'c -> 'b`



Partial Application

```
# (+);;
```

```
- : int -> int -> int = <fun>
```

```
# (+) 2 3;;
```

```
- : int = 5
```

```
# let plus_two = (+) 2;;
```

```
val plus_two : int -> int = <fun>
```

```
# plus_two 7;;
```

```
- : int = 9
```

- Patial application also called *sectioning*



Lambda Lifting

- You must remember the rules for evaluation when you use partial application

```
# let add_two = (+) (print_string "test\n"; 2);;
```

```
test
```

```
val add_two : int -> int = <fun>
```

```
# let add2 = (* lambda lifted *)
```

```
  fun x -> (+) (print_string "test\n"; 2) x;;
```

```
val add2 : int -> int = <fun>
```



Lambda Lifting

```
# thrice add_two 5;;
```

```
- : int = 11
```

```
# thrice add2 5;;
```

```
test
```

```
test
```

```
test
```

```
- : int = 11
```

- Lambda lifting delayed the evaluation of the argument to (+) until the second argument was supplied



Partial Application and “Unknown Types”

- Consider compose plus_two:

```
# let f1 = compose plus_two;;
```

```
val f1 : ('_a -> int) -> '_a -> int = <fun>
```

- Compare to lambda lifted version:

```
# let f2 = fun g -> compose plus_two g;;
```

```
val f2 : ('a -> int) -> 'a -> int = <fun>
```

- What is the difference?



Partial Application and “Unknown Types”

- ‘_a can only be instantiated once for an expression

```
# f1 plus_two;;
```

```
- : int -> int = <fun>
```

```
# f1 List.length;;
```

Characters 3-14:

```
f1 List.length;;
```

```
^^^^^^^^^^^^
```

This expression has type 'a list -> int but is here used
with type int -> int



Partial Application and “Unknown Types”

- ‘a can be repeatedly instantiated

```
# f2 plus_two;;
```

```
- : int -> int = <fun>
```

```
# f2 List.length;;
```

```
- : 'a list -> int = <fun>
```



Continuations

- Idea: Use functions to represent the control flow of a program
- Method: Each procedure takes a function as an argument to which to pass its result; outer procedure “returns” no result
- Function receiving the result called a continuation
- Continuation acts as “accumulator” for work still to be done



Example of Tail Recursion

```
# let rec prod l =  
  match l with [] -> 1  
  | (x :: rem) -> x * prod rem;;  
val prod : int list -> int = <fun>  
# let prod list =  
  let rec prod_aux l acc =  
    match l with [] -> acc  
    | (y :: rest) -> prod_aux rest (acc * y)  
  (* Uses associativity of multiplication *)  
  in prod_aux list 1;;  
val prod : int list -> int = <fun>
```

Example of Tail Recursion

```
# let rec app fl x =  
  match fl with [] -> x  
  | (f :: rem_fs) -> f (app rem_fs x);;  
val app : ('a -> 'a) list -> 'a -> 'a = <fun>  
# let app fs x =  
  let rec app_aux fl acc=  
    match fl with [] -> acc  
    | (f :: rem_fs) -> app_aux rem_fs  
      (fun z -> acc (f z))  
  in app_aux fs (fun y -> y) x;;  
val app : ('a -> 'a) list -> 'a -> 'a = <fun>
```




Continuation Passing Style

- Writing procedures so that they take a continuation to which to give (pass) the result, and return no result, is called continuation passing style (CPS)



Example of Tail Recursion & CSP

```
# let app fs x =  
  let rec app_aux fl acc=  
    match fl with [] -> acc  
    | (f :: rem_fs) -> app_aux rem_fs  
                        (fun z -> acc (f z))  
  in app_aux fs (fun y -> y) x;;  
val app : ('a -> 'a) list -> 'a -> 'a = <fun>  
# let rec appk fl x k =  
  match fl with [] -> k x  
  | (f :: rem_fs) -> appk rem_fs x (fun r -> k (f r));;  
val appk : ('a -> 'a) list -> 'a -> ('a -> 'b) -> 'b
```



Example of CSP

```
# let rec app fl x =  
  match fl with [] -> x  
  | (f :: rem_fs) -> f (app rem_fs x);;  
val app : ('a -> 'a) list -> 'a -> 'a = <fun>
```

```
# let rec appk fl x k =  
  match fl with [] -> k x  
  | (f :: rem_fs) -> appk rem_fs x (fun r -> k (f r));;  
val appk : ('a -> 'a) list -> 'a -> ('a -> 'b) -> 'b =  
  <fun>
```



Continuation Passing Style

- A programming technique for all forms of “non-local” control flow:
 - non-local jumps
 - exceptions
 - general conversion of non-tail calls to tail calls
- Essentially it's a higher-order version of GOTO



Continuation Passing Style

- A compilation technique to implement non-local control flow, especially useful in interpreters.
- A formalization of non-local control flow in denotational semantics (CS 422)



Terms

- A function is in Direct Style when it returns its result back to the caller.
- A Tail Call occurs when a function returns the result of another function call without any more computations (eg tail recursion)
- A function is in Continuation Passing Style when it passes its result to another function.
- Instead of returning the result to the caller, we pass it forward to another function.



Example

- Simple reporting continuation:

```
# let report x = (print_int x; print_newline () );;  
val report : int -> unit = <fun>
```

- Simple function using a continuation:

```
# let plusk a b k = k (a + b)  
val plusk : int -> int -> (int -> 'a) -> 'a = <fun>  
# plusk 20 22 report;;  
42  
- : unit = ()
```